


JE520-CM

VOICE SYNTHESIZER

FOR COMMODORE 64 AND VIC-20



USER'S MANUAL

This manual is intended for users of the Jameco JE520 Series Voice Synthesizer, including users of commercially available software, as well as software developers. These pages describe installation and operation of the JE520 with a Commodore* 64 or VIC-20 personal computer. Programming instructions are given for speech using BASIC and machine language.

Jameco Electronics makes no warranties, expressed or implied, concerning the accuracy of this manual. Every effort is made to insure that the contents are correct but errors may occur.

Preliminary Edition 1-3-84

Copyright 1983 by Jameco Electronics
1355 Shoreway Road
Belmont, CA 94002
(415) 592-8097

Written by David H Nelson - Jameco Electronics

* Commodore 64 & VIC-20 are registered trademarks of Commodore Business Machines, Inc.

TABLE OF CONTENTS

	Introduction	ii
1	USES OF JEF What can JEF do?	1
2	OPERATION How does JEF work?	2
3	BASIC PROGRAMMING How do I use JEF?	5
4	MACHINE LANGUAGE PROGRAMMING How to do it.	10
APPENDICES		
	A - Base Program	12
	B - JEF's Address Book for Commodore 64 and VIC-20	14
	C - Commodore Interface Diagram	15
	D - JEF's Vocabulary List	16
	E - JEF's Extended Vocabulary List	19
	F - JEF's Alphabetical Vocabulary List	20

INTRODUCTION

The JE520 Series Voice Synthesizer is a speech output device for personal computers. It has such a personality that you'll soon understand why we like to call it JEF. He will add a new dimension to your computer that will qualify the name 'personal computer.'

JEF will communicate with you on your own level; human speech. His speech quality rivals that of human speech and is in fact based on a real male voice. The built-in speaker with adjustable volume delivers very clear and audible sound. The all-metal case serves to enhance the sound quality, to eliminate radio interference, and to protect JEF from rough handling.

A subminiature jack allows JEF's audio output to be connected to an external speaker. It may also be used for direct connection to external electronic equipment such as amplifiers, tape recorders, and televisions.

JEF has a basic vocabulary of 268 words. These words include numbers, letters, colors, and many basic words such as prepositions, verbs, and adjectives. Five prefixes and ten suffixes expand the basic vocabulary enormously. Four different tones (beeps) and four different length silences (pauses used to separate words) round out the basic vocabulary. A future expansion unit will allow JEF to be expanded in groups of about 130 words. Complete software support will be provided for all words.

We at Jameco Electronics are dedicated to serving you, the customer, by offering, among other things, this User's Manual and free software with JEF. We will be conducting an on-going program to produce and gather software of all types.

USES OF JEF

JEF has many interesting applications. He will enhance the usefulness of new and existing software of all types. If you have a special use for JEF please let us know what it is. Here are some suggestions.

Education

JEF's applications in education are virtually limitless. He can be used to teach math, english, typing, spelling, reading, pronunciation or any other subject that can be taught with a computer. He can help to teach small children who cannot read or it can drill older children with math problems. He will even improve communication with adults, helping them learn faster.

Games

One of the most exciting uses for JEF is in games. All the excitement of existing games can be expanded to stimulate the player with intelligent voice output. Enhancements to computer games will make them more realistic and entertaining. Imagination is the only limit here, from games as traditional as checkers to those as modern as space games.

Instrumentation

JEF is great for any application where the operator can't take his eyes off another process long enough to read a printed message. Electronic instrumentation can be monitored and results audibly indicated.

Entertainment

JEF can be used for entertainment or just as a conversation piece. He will talk for hours and never tire.

Warning systems

A spoken warning might alert users of danger without their visual attention. This would leave the user free to do other tasks without fear of missing an important message.

Handicap

Being an audio device, JEF is a perfect aide for the visually handicapped. The applications range from warning systems to entertainment, to teaching the blind to type. He offers a new form of feedback from the usual video display of the home computer. He will also allow the mute to communicate vocally.

OPERATION

JEF uses National Semiconductor's Digitaltalker™ speech synthesis chip to produce his realistic vocal output. JEF's four custom 64K EPROMs contain his 268 words.

Also built into JEF's main unit are a speaker, an audio amplifier, an interface to your computer, and circuitry to allow expansion to over 1000 words.

JEF's built-in speaker means that no other components are necessary to allow your computer to talk. The internal amplifier has a control to change the volume. To adjust the output insert a small screwdriver through the hole in the bottom of the case and turn the volume control for a comfortable listening level.

There is a subminiature jack located on JEF's right rear side to allow connection to an external speaker. It may also be used for direct connection to other electronic equipment. Inserting a subminiature plug into this jack will disconnect the internal speaker. The volume control changes the output level at the jack.

JEF's speech EPROMs are divided into groups. Each group is made up of two 64K EPROMs which contain the data for about 130 words. There are two word groups in JEF and he will support up to eight.

Selecting a particular word in JEF's Vocabulary is a two step process. First JEF is told which word group to use. Then he is told which particular word in that group to say. This process will be covered in **BASIC PROGRAMMING**. JEF's Vocabulary List is given in Appendix D.

TIP

Normally the data lines are not changed while JEF is still speaking. Doing so on the 64 will start JEF saying the next word. This can be an interesting technique when used properly. New words may be started slightly before the previous word is finished with little decrease in comprehension. Experimentation will uncover the details of this technique.

Notice several things about the way JEF's Vocabulary List is numbered. First notice that there are no words numbered 0 through 7. Remember that JEF will support up to 8 word groups (0 through 7). These numbers are reserved for selecting the proper group. The first actual word in each group is number 8.

TIP

Selecting a word number that is not on the Vocabulary List will cause JEF to make strange noises. You may find it interesting to try this and can use these sounds for special effects.

Also notice that there is a break in the sequence of numbers. It stops at 156 and starts up again at 264. This break happens when we add 256 times the group number to the base word number. In other words, to get the number of the first word in group 1 we add 256 times 1 to the first word number in that group, 8, and get 264. Only groups 0 and 1 are contained inside JEF. Words numbered 8 through 156 are in group 0. Words numbered 264 through 382 are in group 1 and so on. Groups 2 through 7 start at 520, 776, 1032, 1288, 1544, and 1800 respectively.

JEF is interfaced to your Commodore computer through the parallel user port. He also draws his power from the user port. The exact circuitry is slightly different in the Commodore 64 and the VIC-20 (See Appendix C for more detail). In either case the operation of JEF is the same.

TECHNICAL NOTE

There are 8 parallel data lines and 3 control lines used to control JEF's speech. One control line signals him that the data will select the GROUP. The second line tells him to SPEAK. The remaining line signals the computer that JEF is BUSY talking.

NOTE FOR VIC-20 USERS

The Datasette™ switch sense of the VIC-20 is tied to JEF's SPEAK line. If any of the switches on the Datasette™ are pressed down JEF will refuse to talk when instructed to by a program. Each time a switch is pressed and then released JEF will say the most recently spoken word. This feature might be useful when reviewing words. Be sure none of the Datasette™ switches are down before attempting to use JEF.

Installing JEF is simple. First make sure your Commodore 64 or VIC-20 is off. Installing JEF with the power on may damage him or the computer. Make sure the **TOP** side of the -CM Commodore adapter is up and plug it into the user port. The user port is the connector on the far left side of the computer (figure 1). That's all there is to it. You may leave JEF plugged in all the time but if you decide to remove him for any reason just be sure the power is off before you unplug the connector.

JE520 - CM INSTALLATION




Figure 1

BASIC PROGRAMMING

JEF is easily programmed from any BASIC program. A few simple lines allow him to say any word in his vocabulary. Most people will find it easy to add speech to existing BASIC programs.

Programming

Programming JEF to speak is accomplished in two stages. First the user port is set up. This should be done at the beginning of the program, before any words are spoken. A few POKE statements will take care of this. The last stage is to tell JEF which word he should say. This should be done by GOSUBing to a subroutine that will POKE the correct values to the port.

To get you started type in this program.

Commodore 64

```
10 REM SET UP PORT
20 TK=56576
30 POKE TK+13,16
40 POKE TK+3,255
50 POKE TK+2,PEEK(TK+2) OR 4
60 POKE TK,PEEK(TK) OR 4
70 WD=0:GOSUB 30060

29999 END
30000 REM SAY A WORD
30010 GR=INT(WD/256)
30020 WR=WD-(GR*256)
30030 DL=0
30040 IF(PEEK(TK+13)AND16)=16 THEN 30060
30050 DL=DL+1:IFDL<40 THEN 30040
30060 POKE TK+1,GR
30070 POKE TK,PEEK(TK) AND 251
30080 POKE TK,PEEK(TK) OR 4
30090 DL=PEEK(TK+13)
30100 POKE TK+1,WR
30110 RETURN
```

VIC-20

```
10 REM SET UP PORT
20 TK=37136
30 POKE TK+12,PEEK(TK+
12) AND 15 OR 224
40 POKE TK+2,255
50 POKE TK+3,PEEK(TK+3
) OR 64
60 POKE TK+15,PEEK(TK+
15) OR 64
70 WD=0:GOSUB 30060

29999 END
30000 REM SAY A WORD
30010 GR=INT(WD/256)
30020 WR=WD-(GR*256)
30030 DL=0
30040 IF(PEEK(TK+13)AN
D16)=16 THEN 30060
30050 DL=DL+1:IF DL<40
THEN 30040
30060 POKE TK,GR
30070 POKE TK+12,PEEK(
TK+12) OR 32
30080 POKE TK+12,PEEK(
TK+12) AND 223
30090 DL=PEEK(TK+13)
30100 POKE TK,WR
30102 POKE TK+15,PEEK(
TK+16) AND 191
30104 POKE TK+15,PEEK(
TK+16) OR 64
30110 RETURN
```

These are the Base Programs and will be the starting point for all of our examples. Lines 10 through 70 set up the port. They are only executed once. Lines 30000 through 30110 are the subroutine that says a particular word. Any lines from 100 up to 29999 will be the main body of the program. These lines will be changed to accomplish different tasks.

When you write a new program start with the Base Program. Doing so will insure compatibility of your programs. It will also eliminate the need to rewrite a speech routine each time you write a program. A more detailed description of the base program is given in Appendix A.

TIP

Be sure that JEF is installed before you RUN any speech programs. If he is not, the Base Program will wait about half a second for each word to be spoken. The program will still run but you won't hear any speech.

Enter the following lines.

```
100 FOR WD=8 TO 156
110 GOSUB 30000
120 NEXT WD
130 FOR WD=264 TO 382
140 GOSUB 30000
150 NEXT WD
```

Now RUN the program. JEF says each of the words on the Vocabulary List in the order they are numbered. This program uses the subroutine at line 30000 to say each word. To use this subroutine just set WD to the word number on the Vocabulary List and GOSUB 30000. The complete Vocabulary List can be found in Appendix D.

Subroutine 30000 waits up to half a second for any word in process to finish before saying the next word. Then it selects the word group and finally selects the specific word. This process allows JEF to produce continuous speech. To make the speech more realistic short silences may be placed between words. This can be done by using the four silences provided in JEF's vocabulary. Words 9, 10, 11, and 12 are the silences ranging from 20 ms to 640 ms (1 ms = 1 millisecond = one thousandth of one second). These silences may be combined to make many different duration pauses.

Words starting or ending with 'hard' sounds like B, D, G, K, P, or T need to be separated from adjacent words by a short silence. Try an 80 ms pause before words starting in these sounds. For words ending in these sounds follow them with a 40 ms silence. Silences should also be used at punctuation marks like commas and periods. Trial and error will tell which silences are appropriate.

One point to keep in mind is that when JEF is programmed from BASIC there are already delays between words due to the slowness of BASIC. The base program takes about 20 to 40 ms between each word.

Let's try another example. Enter the following lines to your Base Program and RUN it.

```
100 READ WD$,WD
110 IF WD=0 THEN PRINT:GOTO 29999
120 GOSUB 30000
130 PRINT WD$;
140 GOTO 100
1000 DATA "HELLO",279," ",12,"WELCOME "
,374,"TO ",50,"THE ",356,"JAMECO ",8
1010 DATA "J",74,"E",69,"5",53,"20 ",101
,"VOICE ",370,"SYSTEM.",350,"",0
```

(Don't forget to take out line 150 from the last example)

Notice the pause between 'Hello' and 'Welcome.' It is word 12, the longest silence (640 ms). There are no other silences in the this program. All other pauses are caused by the slowness of BASIC. Most of the time this pause is long enough and no intentional pauses are needed. There are four different length silences to use. Word 9 is the shortest at 20 ms. Word 10 is 40 ms. Word 11 is doubled again to 80 ms. Word 12 is the longest silence at 640 ms. Any of the silences may be used to improve speech quality.

Try changing the DATA statements to say other words. Data should be a word in quotes followed by the word number. You may spell the words any way you like. The last word will be number 0.

To expand JEF's Vocabulary you can use prefixes and suffixes with the other words. There are 10 suffixes numbered 17 to 26 and 5 prefixes from 27 to 31. To make a compound word say the base word directly adjacent to the desired prefix or suffix with no silence between them. This will make the combination sound like one word. You can make nouns plural by adding a form of -s; verbs past tense by adding a form of -ed; and many other vocal tricks.

Add the following lines to your program and RUN it.

```
1000 DATA "MY ",307,"CONTROLLER ",133," "
,19,"IS ",289,"FASTER ",265,"",20
1010 DATA "THAN ",354,"YOURS.",382,"",23
," A ",65,"SPELLING ",347,"",22
1020 DATA "BEE ",66,"SHOULD ",343,"NOT "
,39,"BE ",66,"BLUE.",120,"",0
```

This example shows us how to use prefixes and suffixes. It also shows us how one word can have more than one meaning. Words with the same pronunciation but different meanings and spellings are called homonyms. The word 'BEE' and 'BE' come from the word 'B' listed in JEF's Vocabulary List.

Many of JEF's words can be used for more than one meaning. 'B' can be 'be' or 'bee.' 'C' can be 'see' or 'see.' 'U' can be 'you.' '2' can be 'to,' 'too' or even 'two.' There are hundreds of possibilities. We have made a list of some of the homonyms we have found as well as many compound words. This list is by no means complete but should get you started. It is JEF's Extended Vocabulary List (Appendix E).

Also included in JEF's Vocabulary are four different tones. They can be used as sound effects, warning signals or indicators of any kind. Word 13 is a long (500 ms) 80 Hz tone and 14 is a long (500 ms) 800 Hz tone. Word 15 is a short (50 ms) 1 KHz tone and 16 is a short (50 ms) 2 KHz tone.

Adding Speech to Programs

You may already have a collection of BASIC programs that you would like to enhance with the power of speech. This is an excellent idea and fairly easy to accomplish. From now on your programs will talk to you.

Your first task is the hardest one and no computer will be able to help you with it. You must decide what you want your program to say and where you want to say it. To help you realize all the possibilities, keep the Vocabulary Lists (Appendix D, E, and F) nearby.

Speaking programs might ask the user for an input, instruct them what to do, or warn them of an error. Games can include speech with sound effects and graphics. Write down the words you want to add and find the location in the program where you want to add them. You might place them just before an INPUT statement or in an error handling routine.

Now for the programming. Type the Base Program into the program you are modifying. Change the line numbers if necessary to avoid existing lines.

Add a statement to your program for each word to be said. The statements will look something like this:

```
100 WD=279 : GOSUB 30000
```

They will be placed at the point in the program where you want to say the words. WD is set to the word number to be spoken and then we GOSUB 30000 to the SAY A WORD subroutine. (If you changed the line numbers of the Base Program then you should change the GOSUB to the new line number. When this point is reached in your program the subroutine will wait for any words in process to finish and then instruct JEF to say word WD. It will wait about half a second for a word in process to finish and then assume that JEF is not connected. If you don't want your program to wait for words to finish then take out lines 30030 through 30050. This might be a good idea in fast-paced games but you'll have to be careful not to overlap words.

When you have added all the words to your program RUN it to be sure they are in the correct places. Change them if necessary. When the new program meets your satisfaction save it, possibly under a new name if you want to keep the original version around. The process is complete. From now on you'll be able to listen to your program instead of just watching it.

An impressive technique to try is to PRINT a word as JEF says it. We did this in a previous example. Being able to see and hear the words makes them easier to understand.

Adding speech to programs only makes them slightly longer. You can change all of your programs or just some. Be on the lookout for new applications of speech in future programs. Completely new concepts can be explored. If you have or write programs in machine language that you would like to have speech, go on to the next chapter called . . .

MACHINE LANGUAGE PROGRAMMING

Machine language and assembly language programming (ML) is beyond most JEF users capabilities. It is, however, useful for those who wish to write games or other programs that require faster speed than is offered by BASIC. We will not attempt to teach 6500 microprocessor ML here but only to assist those who are comfortable with it already. Commodore's **Programmers Reference Guides** are an excellent source of information for ML programmers. If you are not familiar with machine language you may want to follow the text of this chapter but skip the programming details.

Making JEF talk with ML is similar to BASIC programming. A ML program will proceed in two stages as the BASIC program did. First it will set up the port and then, as each word is needed, the subroutine will be called to say the word.

The ML program will mimic the BASIC program but will perform much more quickly. As with BASIC, the ML program for the Commodore 64 will be different than for the VIC-20 due to different hardware configurations.

Here is the program to set up the port:

Commodore 64	VIC-20
AD 0D DD LDA \$DD0D	AD 1C 91 LDA \$911C
09 10 ORA #\$10	09 10 ORA #\$0F
8D 0D DD STA \$DD0D	29 E0 AND #\$E0
A9 FF LDA #\$255	8D 1C 91 STA \$911C
8D 03 DD STA \$DD03	A9 FF LDA #\$255
AD 02 DD LDA \$DD02	8D 12 91 STA \$9112
09 04 ORA #\$04	AD 13 91 LDA \$9113
8D 02 DD STA \$DD02	09 04 ORA #\$40
AD 00 DD LDA \$DD00	8D 13 91 STA \$9113
09 04 ORA #\$04	AD 1F 91 LDA \$911F
8D 00 DD STA \$DD00	09 04 ORA #\$40
60 RTS	8D 1F 91 STA \$911F
	60 RTS

The subroutines appear longer than their BASIC counterparts but in fact take up less memory and are hundreds of times faster. The subroutines perform the same function as lines 10 through 70 of the BASIC Base Program. They may be placed anywhere in memory since there are no address dependent instructions. (The tape buffer can be used on both machines and locations C000 through CFFF are available for ML programs on the Commodore 64.) To set up the port, simply JSR to the starting address of your routine. When the subroutine is finished (almost instantly) it will return to the instruction after the JSR.

Now it's time to say a word. Your program will set the Y register to the proper group number and the X register to the word in that group. It will then JSR to the following subroutine which will tell JEF to say the word.

Commodore 64

```
AD 0D DD LDA $DD0D
29 10   AND #$10
F0 F9   BEQ $F9
8C 01 DD STY $DD01
AD 00 DD LDA $DD00
29 FB   AND #$FB
8D 00 DD STA $DD00
09 04   ORA #$04
8D 00 DD STA $DD00
8E 01 DD STX $DD01
60     RTS
```

VIC-20

```
AD 1D 91 LDA $911D
29 10   AND #$10
F0 F9   BEQ $F9
8C 10 91 STY $9110
AD 1C 91 LDA $911C
29 DF   AND #$DF
8D 1C 91 STA $911C
09 20   ORA #$20
8D 1C 91 STA $911C
8E 10 91 STX $9110
AD 1F 91 LDA $911F
29 BF   AND #$BF
8D 1F 91 STA $911F
09 40   ORA #$40
8D 1F 91 STA $911F
60     RTS
```

The first part of the subroutine waits for any word in process to finish. Unlike the BASIC program however, it will wait indefinitely if JEF is not connected. In this case you'll probably have to reset the machine to get it to come back. It is important to be sure that JEF is connected so this will not happen.

There are other ways to program JEF in ML. We have shown you the easiest and most straightforward method. Other techniques include attaching subroutines to processor interrupts.

A modification of the 60 Hz interrupt routine could eliminate the need to wait for a word in process to finish. It could pole the DONE flag and, when it is ready, send the next word.

Another way to eliminate the wait is to use the DONE flag from the BUSY line with an interrupt routine to say the next word. This routine might create a software buffer to hold words that are waiting to be spoken. Another advantage of this approach is that can run in the background (user transparent) without the need to constantly check the DONE flag.

Congratulations! You have entered the realm of computer voice synthesis, a yet untapped resource that will someday improve the quality of all our lives. In the meantime we wish you good luck and hope JEF enlightens your life.

BASE PROGRAM

Commodore 64

```

10 REM SET UP PORT
20 TK=56576
30 POKE TK+13,16
40 POKE TK+3,255
50 POKE TK+2,PEEK(TK+2) OR 4
60 POKE TK,PEEK(TK) OR 4
70 WD=0:GOSUB 30060

29999 END
30000 REM SAY A WORD
30010 GR=INT(WD/256)
30020 WR=WD-(GR*256)
30030 DL=0
30040 IF(PEEK(TK+13)AND16)=16 THEN 30060
30050 DL=DL+1:IFDL<40 THEN 30040
30060 POKE TK+1,GR
30070 POKE TK,PEEK(TK) AND 251
30080 POKE TK,PEEK(TK) OR 4
30090 DL=PEEK(TK+13)
30100 POKE TK+1,WR
30110 RETURN

```

VIC-20

```

10 REM SET UP PORT
20 TK=37136
30 POKE TK+12,PEEK(TK+
  12) AND 15 OR 224
40 POKE TK+2,255
50 POKE TK+3,PEEK(TK+3
  ) OR 64
60 POKE TK+15,PEEK(TK+
  15) OR 64
70 WD=0:GOSUB 30060

29999 END
30000 REM SAY A WORD
30010 GR=INT(WD/256)
30020 WR=WD-(GR*256)
30030 DL=0
30040 IF(PEEK(TK+13)AN
  D16)=16 THEN 30060
30050 DL=DL+1:IF DL<40
  THEN 30040
30060 POKE TK,GR
30070 POKE TK+12,PEEK(
  TK+12) OR 32
30080 POKE TK+12,PEEK(
  TK+12) AND 223
30090 DL=PEEK(TK+13)
30100 POKE TK,WR
30102 POKE TK+15,PEEK(
  TK+16) AND 191
30104 POKE TK+15,PEEK(
  TK+16) OR 64
30110 RETURN

```

Using the Base Program to write new programs is simple. Start by typing it in. Write your new program starting at line 100 up to 29999. When ever you would like to have your program speak, set WD to the word you need and GOSUB 30000. Repeat this process for each word to be spoken. Experimentation will yield the best operation of your final program.

The user port interface of the Commodore 64 is different than the VIC-20's. The 64 uses a 6526 Complex Interface Adapter (CIA) while the VIC-20 uses a 6522 Versatile Interface Adapter (VIA). Although the circuitry is different, the signals are still the same. The programming is only slightly changed.

The following is a detailed breakdown of the Base Program.

Base Program description

LINE(S)	DESC
10-20	Set TK to base address of chip (CIA or VIA)
30	64 - Clear FLAG bit VIC-20 - Set SPEAK high
40	Set DATA0-7 as output
50-60	Set GROUP as ouput and set it low
70	Set FLAG by saying short pause
29999	End of program
30000-30020	Separate word(WD) into group(GR) and word(WR)
30030-30050	Wait 40 loops for word to finish otherwise go on
30060	Put group(GR) on data lines
30070-30080	Latch the group into JEF
30090	Clear FLAG by reading interrupt flag register
30100	Put word(WR) on data lines
30102-30104	(VIC-20 only) SPEAK the word
30110	RETURN from the subroutine

JEF's ADDRESS BOOK FOR COMMODORE 64 AND VIC-20

COMMODORE 64

ADDRESS	BIT(S)	USER PORT NAME	JE520 NAME	DESC
56576	2	PA2	GROUP	PERIPHERAL DATA REG A (BIT 2)
56577	0-7	PB0-7	DATA0-7	PERIPHERAL DATA REG B
56578	2	DDRA		DATA DIRECTION REG A
56579	0-7	DDRB		DATA DIRECTION REG B
56589	4	FLAG2	DONE	INTERRUPT CONTROL REG (FLAG2)


VIC-20

ADDRESS	BIT(S)	USER PORT NAME	JE520 NAME	DESC
37136	0-7	PB0-7	DATA0-7	PERIPHERAL DATA REG B
37138	0-7	DDRB		DATA DIRECTION REG B
37139	6	DDRA		DATA DIRECTION REG A
37148	4	CB1	DONE	PERIPHERAL CONTROL REG
37148	5-7	CB2 CTL	GROUP	CB2 CONTROL
37151	6	PA6	SPEAK	PERIPHERAL DATA REG A

COMMODORE INTERFACE DIAGRAM


COMMODORE 64 USER PORT
CIA* NO. 1 6526

JAMECO JE520-CM
COMMODORE ADAPTER BOARD



VIC-20 USER PORT
VIA** NO. 1 6522

JAMECO JE520-CM
COMMODORE ADAPTER BOARD



* CIA - Complex Interface Adapter

** VIA - Versatile Interface Adapter

JEF'S VOCABULARY LIST

The following is JEF's master Vocabulary List. It includes numbers, letters, prepositions, nouns, verbs, adjectives, silences, tones, etc. Most of the first 128 words (8-127) have significance with their ASCII (American Standard Code for Information Interchange) character counterparts. They are therefore shown with the ASCII characters.

DEC is the master word number (in decimal) of the particular word including the group number. These are used for BASIC programming. Sticking to this format will ensure software compatibility. HEX is the word number (in hexadecimal) disregarding the group number. It can be used for machine language programming. ASC (8-127 only) is the ASCII character that corresponds to that word.

WORD is the name or description (i.e. silences and tones) of the particular word. A dash (-) before or after the word indicates that it is used as a prefix or suffix. They may be used alone but they are meant to blend with other words.

WORD GROUP 0

DEC	HEX	ASC	WORD	DEC	HEX	ASC	WORD
0				32	20	SP	SPACE
1				33	21	!	ATTENTION
2			RESERVED FOR	34	22	"	THANK-YOU
3				35	23	#	NUMBER
4			GROUP SELECTION	36	24	\$	DOLLAR
5				37	25	%	PERCENT
6				38	26	&	AND
7				39	27	'	NOT
8	08	BS	JAMECO	40	28	(OPEN
9	09	HT	20MS silence	41	29)	CLOSE
10	0A	LF	40MS silence	42	2A	*	TIME
11	0B	VT	80MS silence	43	2B	+	PLUS
12	0C	FF	640MS silence	44	2C	,	OR
13	0D	CR	80HZ tone (500 ms)	45	2D	-	MINUS
14	0E	SO	800HZ tone (500 ms)	46	2E	.	POINT
15	0F	SI	1KHZ tone (50 ms)	47	2F	/	DIVIDE
16	10	DLE	2KHZ tone (50 ms)	48	30	0	0 (zero)
17	11	DC1	-ED as 'started'	49	31	1	1 (one)
18	12	DC2	-D as 'closed'	50	32	2	2 (two)
19	13	DC3	-R as 'helper'	51	33	3	3 (three)
20	14	DC4	-ER as 'lesser'	52	34	4	4 (four)
21	15	NAK	-NG as 'missing'	53	35	5	5 (five)
22	16	SYN	-ING as 'inserting'	54	36	6	6 (six)
23	17	ETB	-S as 'points'	55	37	7	7 (seven)
24	18	CAN	-ES as 'spaces'	56	38	8	8 (eight)
25	19	EM	-TH as 'fourth'	57	39	9	9 (nine)
26	1A	SUB	-ETH as 'twentieth'	58	3A	:	LINE
27	1B	ESC	A- as 'across'	59	3B	;	CONTINUE
28	1C	FS	IN- as 'inset'	60	3C	<	LESS
29	1D	GS	R- as 'replace'	61	3D	=	EQUAL
30	1E	RS	RE- as 'reset'	62	3E	>	GREAT
31	1F	US	UN- as 'unclear'	63	3F	?	HELP

WORD GROUP 0 (cont)

DEC	HEX	ASC	WORD	DEC	HEX	ASC	WORD
64	40	@	AT	96	60	`	15 (fifteen)
65	41	A	A	97	61	a	16 (sixteen)
66	42	B	B	98	62	b	17 (seventeen)
67	43	C	C	99	63	c	18 (eighteen)
68	44	D	D	100	64	d	19 (nineteen)
69	45	E	E	101	65	e	20 (twenty)
70	46	F	F	102	66	f	30 (thirty)
71	47	G	G	103	67	g	40 (forty)
72	48	H	H	104	68	h	50 (fifty)
73	49	I	I	105	69	i	60 (sixty)
74	4A	J	J	106	6A	j	70 (seventy)
75	4B	K	K	107	6B	k	80 (eighty)
76	4C	L	L	108	6C	l	90 (ninety)
77	4D	M	M	109	6D	m	AFTER
78	4E	N	N	110	6E	n	AGAIN
79	4F	O	O	111	6F	o	ALL
80	50	P	P	112	70	p	AN
81	51	Q	Q	113	71	q	AWAY
82	52	R	R	114	72	r	BACK
83	53	S	S	115	73	s	BAR
84	54	T	T	116	74	t	BEEN
85	55	U	U	117	75	u	BEFORE
86	56	V	V	118	76	v	BETWEEN
87	57	W	W	119	77	w	BLACK
88	58	X	X	120	78	x	BLUE
89	59	Y	Y	121	79	y	BREAK
90	5A	Z	Z	122	7A	z	BROWN
91	5B	[10 (ten)	123	7B	}	BY
92	5C	\	11 (eleven)	124	7C		CANCEL
93	5D]	12 (twelve)	125	7D	}	CAUTION
94	5E	^	13 (thirteen)	126	7E	~	CENT
95	5F	_	14 (fourteen)	127	7F	DEL	CHANGE

DEC	HEX	WORD	DEC	HEX	WORD
128	80	CHECK	143	8F	DELETE
129	81	CLEAR	144	90	DEVICE
130	82	COMMAND	145	91	DID
131	83	COMPLETE	146	92	DISK
132	84	CONDITION	147	93	DISPLAY
133	85	CONTROL	148	94	DON'T
134	86	CORRECT	149	95	DOOR
135	87	COST	150	96	DOWN
136	88	CROSS	151	97	DRIVE
137	89	CURSOR	152	98	END
138	8A	CYAN	153	99	ENTER
139	8B	DARK	154	9A	ERROR
140	8C	DATA	155	9B	ESCAPE
141	8D	DAY	156	9C	EXIT
142	8E	DECREASE			

WORD GROUP 1

DEC	HEX	WORD	DEC	HEX	WORD	DEC	HEX	WORD
264	08	FAIL	304	30	MISS	344	58	SIGN
265	09	FAST	305	31	MORE	345	50	SLOW
266	0A	FIRE	306	32	MOVE	346	5A	SPEED
267	0B	FIRST	307	33	MY	347	5B	SPELL
268	0C	FORWARD	308	34	NEXT	348	5C	START
269	0D	FROM	309	35	NO	349	5D	STOP
270	0E	FUEL	310	36	NORMAL	350	5E	SYSTEM
271	0F	FUNCTION	311	37	NOW	351	5F	TAPE
272	10	GET	312	38	O'CLOCK	352	60	TARGET
273	11	GO	313	39	OF	353	61	TEST
274	12	GOOD	314	3A	OFF	354	62	THAN
275	13	GOODBYE	315	3B	OKAY (O.K.)	355	63	THAT
276	14	GREEN	316	3C	ON	356	64	THE
277	15	GREY	317	3D	ORANGE	357	65	THE- (short)
278	16	HAVE	318	3E	OTHER	358	66	THEE- as 'thee'
279	17	HELLO	319	3F	OUT	359	67	THEN
280	18	HERE	320	40	OVER	360	68	THERE
281	19	HIGH	321	41	PHONE	361	69	THIS
282	1A	HIT	322	42	PLACE	362	6A	THOUSAND
283	1B	HOUR	323	43	PLAY	363	6B	TOTAL
284	1C	HUNDRED	324	44	PLEASE	364	6C	TRY
285	1D	IF	325	45	PRESS	365	6D	TURN
286	1E	IN	326	46	PROGRAM	366	6E	TYPE
287	1F	INCREASE	327	47	RANGE	367	6F	UP
288	20	INSERT	328	48	READ (verb)	368	70	USE
289	21	IS	329	49	READY	369	71	VIOLET
290	22	IT	330	4A	RECORD (verb)	370	72	VOICE
291	23	JOYSTICK	331	4B	RED	371	73	WAIT
292	24	KEY	332	4C	REPEAT	372	74	WAS
293	25	LAST	333	4D	REVERSE	373	75	WEAR
294	26	LEFT	334	4E	RIGHT	374	76	WELCOME
295	27	LEVEL	335	4F	ROOM	375	77	WHAT
296	28	LIGHT	336	50	RUN	376	78	WHITE
297	29	LIMIT	337	51	SAVE	377	79	WILL
298	2A	LISTEN	338	52	SECOND	378	7A	WITH
299	2B	LOAD	339	53	SELECT	379	7B	WORD
300	2C	LOW	340	54	SEND	380	7C	YELLOW
301	2D	MAGENTA	341	55	SET	381	7D	YES
302	2E	MILLION	342	56	SHIFT	382	7E	YOUR
303	2F	MINUTE	343	57	SHOULD			

JEF'S EXTENDED VOCABULARY LIST

Besides JEF's many multi-definition words there are also hundreds of other words that can be created from them. Homonyms (words that sound the same but are spelled differently and have different meanings) are abundant in JEF's vocabulary. Other words are created by combining existing words with prefixes and suffixes. Still other words may be created from combinations of words that simply sound like the word needed.

This is a list of some of the "extended" words that we have discovered. It is a small sample but should give you a starting point in your search for new words.

WORD	NUMBERS	WORD	NUMBERS	WORD	NUMBERS
ACROSS	27,136	INCOMPLETE	28,131	RETURN	29,365
ARE	82	INCORRECT	28,134	RUNNER	336,20
ARRANGE	27,327	INN	286	SCENT	126
ATE	56	INSERTING	288,22	SEA	67
BE	66	INSET	28,341	SECONDS	338,23
BEE	66	KNOT	39	SEE	67
BIN	116	KNOW	309	SENSE	126,23
BLEW	120	LEFTOVER	294,320	SENT	126
BRAKE	121	LESSER	60,20	SETTING	341,21
BUY	123	LIGHT BLUE	296,120	SIN	344
BYE	123	LIGHTER	296,19	SINCE	126,23
CENTS	126,23	LOWER	300,19	SIXTH	54,25
CLOSED	41,18	MILLIONTH	302,25	SLOE	345
CLOTHES	41	MINUTES	303,23	SPACES	32,24
CONDITIONER	132,19	MISSED	304,18	SPEEDING	346,21
CONTROLLER	133,19	MISSING	304,21	STARTED	348,17
CUE	81	MRS.	304,24	TEA	84
DARK GREEN	139,276	NO ONE	309,49	TEE	84
DARKER	139,19	OAR	44	THEIR	360
DEVICES	144,24	OH	79	TIMES	42,23
DISK DRIVE	146,151	ONE	49	TO	50
DIVIDED BY	47,17,123	ORE	44	TODAY	50,141
DOLLARS	36,23	OUR	283	TODAY'S	50,141,23
DRIVER	151,19	OWE	79	TOO	50
EQUALS	61,23	PEA	80	TWO	50
EX-	88	PHONE NUMBER	321,35	TWENTIETH	101,26
EXCHANGE	88,127	PLAYER	323,20	UNCLEAR	31,129
EYE	73	PLEASING	324,21	UNLOAD	31,299
FASTER	265,20	POINTER	46,19	UPPER	367,20
FOR	52	POINTS	46,23	USED	368,18
FORE	52	QUEUE	81	WAITING	371,21
GRATE	62	READ	331	WARE	373
GRAY	277	REED	328	WEIGHT	371
GREATER	62,20	REFUEL	30,270	WHERE	373
HEAR	280	RELOAD	30,299	WHY	89
HELPER	63,19	REMOVE	29,306	WITHIN	378,286
HI	281	RENUMBER	30,35	WON	49
HIGHER	281,20	REPLACE	29,322	WRITE	334
HUNDREDS	284,23	REPLAY	30,323	YOU	85
HUNDREDTH	284,25	RESET	30,341	YOU'RE	382

JEF'S ALPHABETICAL VOCABULARY LIST

Word	#	Word	#	Word	#	Word	#	Word	#
-D	18	ATTENTION	33	F	70	MINUTE	303	SHIFT	342
-ED	17	AWAY	113	FAIL	264	MISS	304	SHOULD	343
-ER	20	B	66	FAST	265	MORE	305	SIGN	344
-ES	24	BACK	114	FIRE	266	MOVE	306	SLOW	345
-ETH	26	BAR	115	FIRST	267	MY	307	SPACE	32
-ING	22	BEEN	116	FORWARD	268	N	78	SPEED	346
-NG	21	BEFORE	117	FROM	269	NEXT	308	SPELL	347
-R	19	BETWEEN	118	FUEL	270	NO	309	START	348
-S	23	BLACK	119	FUNCTION	271	NORMAL	310	STOP	349
-TH	25	BLUE	120	G	71	NOT	39	SYSTEM	350
0	48	BREAK	121	GET	272	NOW	311	T	84
1	49	BROWN	122	GO	273	NUMBER	35	TAPE	351
10	91	BY	123	GOOD	274	O	79	TARGET	352
11	92	C	67	GOODBYE	275	O'CLOCK	312	TEST	353
12	93	CANCEL	124	GREAT	62	OF	313	THAN	354
13	94	CAUTION	125	GREEN	276	OFF	314	THANK-YOU	34
14	95	CENT	126	GREY	277	OKAY	315	THAT	355
15	96	CHANGE	127	H	72	ON	316	THE	356
16	97	CHECK	128	HAVE	278	OPEN	40	THE-	357
17	98	CLEAR	129	HELLO	279	OR	44	THEE-	358
18	99	CLOSE	41	HELP	63	ORANGE	317	THEN	359
19	100	COMMAND	130	HERE	280	OTHER	318	THERE	360
1KHZ	15	COMPLETE	131	HIGH	281	OUT	319	THIS	361
2	50	CONDITION	132	HIT	282	OVER	320	THOUSAND	362
20	101	CONTINUE	59	HOUR	283	P	80	TIME	42
20MS	9	CONTROL	133	HUNDRED	284	PERCENT	37	TOTAL	363
2KHZ	16	CORRECT	134	I	73	PHONE	321	TRY	364
3	51	COST	135	IF	285	PLACE	322	TURN	365
30	102	CROSS	136	IN	286	PLAY	323	TYPE	366
4	52	CURSOR	137	IN-	28	PLEASE	324	U	85
40	103	CYAN	138	INCREASE	287	PLUS	43	UN-	31
40MS	10	D	68	INSERT	288	POINT	46	UP	367
5	53	DARK	139	IS	289	PRESS	325	USE	368
50	104	DATA	140	IT	290	PROGRAM	326	V	86
6	54	DAY	141	J	74	Q	81	VIOLET	369
60	105	DECREASE	142	JAMECO	8	R	82	VOICE	370
640MS	12	DELETE	143	JOYSTICK	291	R-	29	W	87
7	55	DEVICE	144	K	75	RANGE	327	WAIT	371
70	106	DID	145	KEY	292	RE-	30	WAS	372
8	56	DISK	146	L	76	READ	328	WEAR	373
80	107	DISPLAY	147	LAST	293	READY	329	WELCOME	374
800HZ	14	DIVIDE	47	LEFT	294	RECORD	330	WHAT	375
80HZ	13	DOLLAR	36	LESS	60	RED	331	WHITE	376
80MS	11	DONT	148	LEVEL	295	REPEAT	332	WILL	377
9	57	DOOR	149	LIGHT	296	REVERSE	333	WITH	378
90	108	DOWN	150	LIMIT	297	RIGHT	334	WORD	379
A	65	DRIVE	151	LINE	58	ROOM	335	X	88
A-	27	E	69	LISTEN	298	RUN	336	Y	89
AFTER	109	END	152	LOAD	299	S	83	YELLOW	380
AGAIN	110	ENTER	153	LOW	300	SAVE	337	YES	381
ALL	111	EQUAL	61	M	77	SECOND	338	YOUR	382
AN	112	ERROR	154	MAGENTA	301	SELECT	339	Z	90
AND	38	ESCAPE	155	MILLION	302	SEND	340		
AT	64	EXIT	156	MINUS	45	SET	341		

REFERENCES

Commodore 64 User's Guide, Commodore Business Machines, Inc.

Commodore 64 Programmer's Reference Guide, Commodore Business Machines, Inc.

VIC Revealed, Hampshire, Nick; Hayden Book Company, Inc.

DIGITALKER Vocabulary Selection System, Speech Synthesis Group,
National Semiconductor Corporation

NOTICE - Changes are periodically made to the information contained herein. These changes will be incorporated into future editions of this manual.

LIMITED WARRANTY

Jameco Electronics warrants the equipment to be free from defects for a period of 6 months from date of purchase. During the warranty the seller agrees to replace or repair the equipment to its normal operating condition, furnishing necessary parts and labor free of charges for the first 90 days. For the remainder of the warranty, after the first 90 days, the equipment is warranted against manufacturer defects on electric components only. The liability shall be limited to replacing or repairing and excludes equipment purchases in kit form. Equipment or parts which have been subject to abuse, misuse, accident, alteration, neglect and/or unauthorized repair are not covered by the warranty. Seller shall have the right of final determination as to existence and cause of defect.

FOR SERVICE, SEND EQUIPMENT TO:

JAMECO ELECTRONICS REPAIR CENTER, 1355 SHOREWAY ROAD, BELMONT, CALIFORNIA 94002. PLEASE PACK SECURELY. WE SUGGEST YOU INSURE THE SHIPMENT.